```
> restart;
> eq:=D(y)=(s->f(s,y(s)));
```

$$eq := \mathrm{D}(y) = (s \mapsto f(s, y(s))) \tag{1}$$

```
> yp:=y(xn)+h*f(xn,y(xn));
```

$$yp := y(xn) + h\, f(xn, y(xn)) \tag{2}$$

```
> ynp1:=y(xn)+h/2*(f(xn,y(xn))+f(xn+h,yp));
```

$$ynp1 := y(xn) + \frac{h\,(f(xn, y(xn)) + f(xn+h, y(xn) + h\, f(xn, y(xn))))}{2} \tag{3}$$

```
> r:=y(xn+h)-ynp1;
```

$$r := y(xn+h) - y(xn) - \frac{h\,(f(xn, y(xn)) + f(xn+h, y(xn) + h\, f(xn, y(xn))))}{2} \tag{4}$$

```
> subs(h=0,r);
```

$$0 \tag{5}$$

```
> T1:=diff(r,h);
```

$$T1 := \mathrm{D}(y)(xn+h) - \frac{f(xn, y(xn))}{2} - \frac{f(xn+h, y(xn) + h\, f(xn, y(xn)))}{2} - \frac{1}{2}\Big(h\,\big(\mathrm{D}_1(f)(xn+h, y(xn) + h\, f(xn, y(xn))) + \mathrm{D}_2(f)(xn+h, y(xn) + h\, f(xn, y(xn)))\, f(xn, y(xn))\big)\Big) \tag{6}$$

```
> dr:=eval(subs(eq,T1));
```

$$dr := f(xn+h, y(xn+h)) - \frac{f(xn, y(xn))}{2} - \frac{f(xn+h, y(xn) + h\, f(xn, y(xn)))}{2} - \frac{1}{2}\Big(h\,\big(\mathrm{D}_1(f)(xn+h, y(xn) + h\, f(xn, y(xn))) + \mathrm{D}_2(f)(xn+h, y(xn) + h\, f(xn, y(xn)))\, f(xn, y(xn))\big)\Big) \tag{7}$$

```
> subs(h=0,dr);
```

$$0 \tag{8}$$

```
> ddr:=eval(subs(eq,diff(dr,h)));
```

$$ddr := \mathrm{D}_1(f)(xn+h, y(xn+h)) + \mathrm{D}_2(f)(xn+h, y(xn+h))\, f(xn+h, y(xn+h)) - \mathrm{D}_1(f)(xn+h, y(xn) + h\, f(xn, y(xn))) - \mathrm{D}_2(f)(xn+h, y(xn) + h\, f(xn, y(xn)))\, f(xn, y(xn)) - \frac{1}{2}\Big(h\,\big(\mathrm{D}_{1,1}(f)(xn+h, y(xn) + h\, f(xn, y(xn))) + \mathrm{D}_{1,2}(f)(xn+h, y(xn) + h\, f(xn, y(xn)))\, f(xn, y(xn)) + \big(\mathrm{D}_{1,2}(f)(xn+h, y(xn) + h\, f(xn, y(xn))) + \mathrm{D}_{2,2}(f)(xn+h, y(xn) + h\, f(xn, y(xn)))\, f(xn, y(xn))\big)\, f(xn, y(xn))\big)\Big) \tag{9}$$

```
> subs(h=0,ddr);
```

$$0 \tag{10}$$

```
> d3r:=eval(subs(eq,diff(ddr,h))):
> simplify(subs(h=0,d3r));
```

$$-\frac{\mathrm{D}_{1,1}(f)(xn, y(xn))}{2} - \mathrm{D}_{1,2}(f)(xn, y(xn))\, f(xn, y(xn)) - \frac{f(xn, y(xn))^2\, \mathrm{D}_{2,2}(f)(xn, y(xn))}{2} + f(xn, y(xn))\, \mathrm{D}_2(f)(xn, y(xn))^2 \tag{11}$$

$$+ D_2(f)(xn, y(xn))\, D_1(f)(xn, y(xn))$$

```
> d4r:=simplify(eval(subs(eq,diff(d3r,h)))):
> simplify(subs(h=0,d4r));
```

$$-f(xn, y(xn))^3 D_{2,2,2}(f)(xn, y(xn)) + \big(4 D_{2,2}(f)(xn, y(xn))\, D_2(f)(xn, y(xn)) - 3 D_{1,2,2}(f)(xn, y(xn))\big) f(xn, y(xn))^2 + \big(D_2(f)(xn, y(xn))^3 + 3 D_{2,2}(f)(xn, y(xn))\, D_1(f)(xn, y(xn)) + 5 D_{1,2}(f)(xn, y(xn))\, D_2(f)(xn, y(xn)) - 3 D_{1,1,2}(f)(xn, y(xn))\big) f(xn, y(xn)) + D_1(f)(xn, y(xn))\, D_2(f)(xn, y(xn))^2 + 3 D_{1,2}(f)(xn, y(xn))\, D_1(f)(xn, y(xn)) + D_2(f)(xn, y(xn))\, D_{1,1}(f)(xn, y(xn)) - D_{1,1,1}(f)(xn, y(xn)) \quad \textbf{(12)}$$

```
> f:=(xi,eta)->A*eta;
```

$$f := (\xi, \eta) \mapsto A\eta \quad \textbf{(13)}$$

```
> method:=y(xn+h)=ynp1;
```

$$method := y(xn+h) = y(xn) + \frac{h\,(A\,y(xn) + A\,(y(xn) + h\,A\,y(xn)))}{2} \quad \textbf{(14)}$$

```
> ceq:=eval(subs(y=(s->rho^s),method));
```

$$ceq := \rho^{xn+h} = \rho^{xn} + \frac{h\left(A\rho^{xn} + A\left(\rho^{xn} + h\,A\,\rho^{xn}\right)\right)}{2} \quad \textbf{(15)}$$

```
> solve(ceq,rho);
```

$$RootOf\left(A^2\, \_Z^{xn}\, h^2 + 2\,h\,A\,\_Z^{xn} - 2\,\_Z^{xn+h} + 2\,\_Z^{xn}\right),\ 0 \quad \textbf{(16)}$$

```
> ceq2:=subs({xn=0,h=1},ceq);
```

$$ceq2 := \rho = 1 + \frac{A}{2} + \frac{A\,(1+A)}{2} \quad \textbf{(17)}$$

```
> S:=solve(ceq2,rho);
```

$$S := 1 + A + \frac{1}{2}A^2 \quad \textbf{(18)}$$

```
> # the linear stability region is all values of A such that
  |rho|<1
> Z1:=subs(A=a+I*b,abs(S));
```

$$Z1 := \left|1 + a + \mathrm{I}\,b + \frac{(a + \mathrm{I}\,b)^2}{2}\right| \quad \textbf{(19)}$$

```
> with(plots):
> contourplot(Z1,a=-2..2,b=-2..2,contours=[1,.9,.8],grid=[100,100])
  ;
```




```
> contourplot(Z1,a=-3..1,b=-2..2,contours=[1],grid=[100,100],
  filled=true);
```




```
> R:=S/exp(A);
```

$$R := \frac{1 + A + \frac{1}{2}A^2}{e^A} \quad (20)$$